
Cykly



Cykly typu WHILE

Cyklus v programu implementujeme skokem nahoru (viz. cv05):
• U cyklu typu while-do se na začátku ptáme jestli se cyklus má ukončit.

1 .while:
2 cmp eax, 0 ; porovnanim hodnot nastavime priznaky
3 jne .end ; podminkou rozhodujeme o ukonceni cyklu
4 .do:
5 ;telo cyklu
6 jmp .while ; ne-podminenym skokem se vracime na test podminky
7 .end:

• U cyklu typu do-while se na konci ptáme jestli se cyklus má opakovat.

8 .do:
9 ;telo cyklu

10 .while:
11 cmp eax, 0 ; porovnanim hodnot nastavime priznaky
12 je .do ; podminkou rozhodujeme o opakovani cyklu
13 .end:

Na vyzkoušení - WHILE-DO

Program bude načítat 32b čísla dokud nezadáme nulu, a pak vypíše jejich sumu:

%include 'rw32.inc' ;knihovna pro vstup a vystup

section .text ;kodovy segment
main:

mov ebx, 0 ; EBX = suma
call ReadInt32NewLine ; EAX = vstup

.while:
cmp eax, 0 ; porovnanim nastavime priznaky
je .end ; pokud EAX == 0 tak cyklus ukoncime

.do: ; jinak se cyklus provede
add ebx, eax ; suma = suma + vstup
call ReadInt32NewLine ; EAX = vstup
jmp .while ; vracime se na zacatek cyklu

.end:

mov eax, ebx ; EAX = suma
call WriteInt32NewLine ; vypis EAX
ret

Na vyzkoušení - DO-WHILE

Program bude načítat 32b čísla dokud nezadáme nulu, a pak vypíše jejich sumu:

%include 'rw32.inc' ;knihovna pro vstup a vystup

section .text ;kodovy segment
main:

mov ebx, 0 ; EBX = suma

.do:
call ReadInt32NewLine ; EAX = vstup
add ebx, eax ; suma = suma + vstup

.while:

cmp eax, 0 ; porovnanim nastavime priznaky
jne .do ; pokud EAX != 0 tak opakujeme cyklus

.end: ; jinak se cyklus ukonci

mov eax, ebx ; EAX = suma
call WriteInt32NewLine ; vypis EAX
ret

Na procvičení

Vyzkoušejte si:
• Ze vstupu si načtěte dvě 8b znaménková čísla.
• Vypište všechny hodnoty v rozsahu od menšího do většího z nich (včetně).
• Jak by se programmusel změnit kdybychom okrajové hodnoty vypisovat nechtěli?

Například:
• 5, 0 => 0 1 2 3 4 5

5, 5 => 5
-5, 5 => -5 -4 -3 -2 -1 0 1 2 3 4 5

Nápověda:
• Podmínkou to menší z obou načtených čísel nastavíme do AL, větší do BL.
• Cyklem vypisujeme a inkrementujeme AL, tak dlouho dokud není větší než BL.
• Pozor na rozdíl mezi znaménkovými a bez-znaménkovými skoky (viz. cv05).

Cyklus typu FOR

Cyklus typu for implementujeme instrukcí LOOP (smyčka):
• Instrukce nejprve dekrementuje (DEC) registr ECX, a pokud jeho
hodnota není rovna nule, tak provede skok na uvedené návěští (JNZ).

Vyzkoušejte si:
• Program pomocí cyklu for (smyčky) vypíše čísla od 10 do 1 (včetně):

%include 'rw32.inc' ;knihovna pro vstup a vystup

section .text ;kodovy segment
main:

mov ecx, 10 ; ECX = pocet opakovani
.for:

mov eax, ecx ; EAX = ECX
call WriteInt32NewLine ; vypis EAX
loop .for ; dekrementuj ECX, a pokud ECX != 0

; tak skoc na navesti .FOR
ret

Průchod polem



Adresování paměti
Při adresování paměti (viz. cv03), obsah hranatých závorek musí pasovat na rovnici
pro výpočet efektivní adresy v 32b chráněném režimu (viz. p3s24):
• “Displacement” je jméno proměnné nebo pole, posunuté o konstantu.
• Registr musí mít 32b, může být násoben konstantou 1/2/4/8, ale nejde odečítat.

Například – správně zapsané efektivní adresy:
1 [arr] ; promenna
2 [arr + 4] ; promenna + posuv
3 [arr + ecx] ; promenna + registr
4 [arr + ecx - 1] ; promenna + registr + (zaporny) posuv
5 [arr + ecx*4 - 4] ; promenna + registr * konstanta + (zaporny) posuv
6 [esi] ; registr
7 [esi + ecx*4 - 4] ; registr + registr * konstanta + (zaporny) posuv

Adresování výpočtem adresy
Při průchodu polem adresu položek můžeme počítat z hodnoty registru ECX:
• Protože se hodnota ECX ve smyčce snižuje, tak polem procházíme z prava do leva.
• Protože velikost offsetu se vždy udává v počtu bajtů, ne v počtu položek,
tak hodnotu registru ECX musíme násobit velikostí položek (1/2/4/8).

• Protože ECX je v rozsahu <1, n>, ne <0, n-1>, tak odčítáme velikost jedné položky.

Vyzkoušejte si:
• Cyklus z prava do leva projde přes všechny položky pole a vypíše je.

%include 'rw32.inc' ;knihovna pro vstup a vystup
section .data ;inicializovany segment

arr dd 100,200,300,400,500; pole peti 32b polozek jmenem arr
section .text ;kodovy segment
main:

mov ecx, 5 ; pocet opakovani cyklu
.for:

mov eax, [arr + ecx*4-4] ; EAX = hodnota aktualniho prvku
; "ECX*4-4" protoze 32 bitu jsou 4 bajty

call WriteInt32NewLine ; vypis EAX
loop .for ; dekrementuj ECX, a pokud ECX != 0 tak skoc
ret

Adresování indexovými registry
Adresu položky můžeme nastavit do jednoho z indexových registrů ESI nebo EDI,
apři průchodu polem pak zvyšovat nebo snižovat jejich hodnotu.
• Adresa první položky je vždy shodná s adresou pole.
• Adresa poslední položky je v paměti posunutá o “(n-1) * velikost položek” bajtů.
• K registru ESI nebo EDI v cyklu přičítáme nebo odčítáme velikost jedné položky.

Vyzkoušejte si:
• Cyklus z leva do prava projde přes všechny položky pole a vypíše je.

%include 'rw32.inc' ;knihovna pro vstup a vystup
section .data ;inicializovany segment

arr dd 100,200,300,400,500; pole peti 32b polozek jmenem arr
section .text ;kodovy segment
main:

mov esi, arr ; adresa prvni polozky pole
mov ecx, 5 ; pocet opakovani cyklu

.for:
mov eax, [esi] ; EAX = hodnota aktualniho prvku
add esi, 4 ; "ESI + 4" protoze 32 bitu jsou 4 bajty
call WriteInt32NewLine ; vypis EAX
loop .for ; dekrementuj ECX, a pokud ECX != 0 tak skoc
ret

Na procvičení

Vyzkoušejte si:
• Vytvořte si inicializované pole pěti 16b bez-znaménkových čísel.
• Projděte přes všechny prvky pole a vypište jejich maximum.

Například:
• 10, 30, 20, 50, 40 => 50
• 10000, 30000, 20000, 50000, 40000 => 50000

Nápověda:
• První nebo poslední z položek pole prohlásíme za maximum, a uložíme ji do AX.
• Cyklem (smyčkou LOOP) projdeme přes všechny položky pole.
• Podmínkou porovnáváme hodnotu AX s hodnotou z aktuální položky pole,
a pokud je AX menší, tak do něj uložíme hodnotu aktuální položky.

• Pozor na rozdíl mezi znaménkovými a bez-znaménkovými skoky (viz. cv05).

ISU-cv07 12 / 30

http://www.fit.vutbr.cz/~ihusa/ISU-cv05.pdf#view=fit


Řetězové instrukce



Řetězové instrukce

Práci s poli (vektory, řetězce) si můžeme usnadnit řetězovými instrukcemi:
• Vždy pracují s registrem AL/AX/EAX (8b, 16b, 32b) a indexovými registry
ESI a EDI, které automaticky posouvají na další položku pole.

• Na rozdíl od ostatních instrukcí, velikost položek a posuvu u nich neurčujeme
pseudo-instrukcemi (byte/word/dword), ale jednou z koncovek B/W/D.

• Umožňují obejít omezení sběrnice (viz. cv03) a pracovat se dvěma hodnotami
uloženými v paměti.

Položky polí můžeme kopírovat (MOVS, LODS, STOS) nebo porovnávat (CMPS, SCAS):

Řetězové kopírování - MOVS, LODS a STOS
Z pole do pole kopírujeme instrukcemi MOVSB, MOVSW, MOVSD (MOVe String):
• Instrukce zkopíruje hodnotu z adresy v ESI na adresu v EDI
a adresy v ESI i v EDI posune na další položku.

1 movsb ;zkopiruj 8b z adresy v ESI na adresu v EDI (a oba posun o 1B)
2 movsw ;zkopiruj 16b z adresy v ESI na adresu v EDI (a oba posun o 2B)
3 movsd ;zkopiruj 32b z adresy v ESI na adresu v EDI (a oba posun o 4B)

Z pole do registru kopírujeme instrukcemi LODSB, LODSW, LODSD (LOaD String):
• Instrukce zkopíruje hodnotu z adresy v ESI do registru AL/AX/EAX
a adresu v ESI posune na další položku.

4 lodsb ;zkopiruj 8b z adresy v ESI do AL (a ESI posun o 1B)
5 lodsw ;zkopiruj 16b z adresy v ESI do AX (a ESI posun o 2B)
6 lodsd ;zkopiruj 32b z adresy v ESI do EAX (a ESI posun o 4B)

Z registru do pole kopírujeme instrukcemi STOSB, STOSW, STOSD (STOre String):
• Instrukce zkopíruje hodnotu z registru AL/AX/EAX na adresu v EDI
a adresu v EDI posune na další položku.

7 stosb ;zkopiruj 8b z AL na adresu v EDI (a EDI posun o 1B)
8 stosw ;zkopiruj 16b z AX na adresu v EDI (a EDI posun o 2B)
9 stosd ;zkopiruj 32b z EAX na adresu v EDI (a EDI posun o 4B)

Na vyzkoušení

Program zkopíruje pět (ECX) 8b položek z pole src (ESI) do pole dst (EDI):

%include 'rw32.inc' ;knihovna pro vstup a vystup
section .data ;inicializovany segment

src db 10,20,30,40,50 ; zdrojove pole peti 8b hodnot
dst db 0, 0, 0, 0, 0 ; cilove pole peti 8b hodnot

section .text ;kodovy segment
main:

mov esi, src ; ESI = adresa zdrojoveho pole
mov edi, dst ; EDI = adresa ciloveho pole
mov ecx, 5 ; ECX = pocet opakovani (pocet polozek pole)
cld ; DF = 0 (posun zleva dodoprava)

.for:
movsb ; zkopiruj 8b z adresy v ESI na adresu v EDI a

; ESI i EDI zvys (posun doprava) o jednicku (8b)

loop .for ; dekrementuj ECX, a pokud ECX != 0 tak skoc
ret

Na vyzkoušení

Program zkopíruje pět (ECX) 8b položek z pole src (ESI) do AL, a z AL do pole dst (EDI):

%include 'rw32.inc' ;knihovna pro vstup a vystup
section .data ;inicializovany segment

src db 10,20,30,40,50 ; zdrojove pole peti 8b hodnot
dst db 0, 0, 0, 0, 0 ; cilove pole peti 8b hodnot

section .text ;kodovy segment
main:

mov esi, src ; ESI = adresa zdrojoveho pole
mov edi, dst ; EDI = adresa ciloveho pole
mov ecx, 5 ; ECX = pocet opakovani (pocet polozek pole)
cld ; DF = 0 (posun zleva dodoprava)

.for:
lodsb ; zkopiruj 8b z adresy v ESI do AL a

; ESI zvys (posun doprava) o jednicku (8b)
stosb ; zkopiruj 8b z AL na adresu v EDI a

; EDI zvys (posun doprava) o jednicku (8b)
loop .for ; dekrementuj ECX, a pokud ECX != 0 tak skoc
ret

Na procvičení

Vyzkoušejte si:
• Vytvořte si inicializované pole pěti 16b znaménkových čísel.
• Hodnotu všech položek pole vydělte deseti, a výsledný obsah
pole si zobrazte v debuggeru.

Například:
• 10, 20, 30, 40, 50 => 1, 2, 3, 4, 5
• 10000, 20000, 30000, -10000, -20000 => 1000, 2000, 3000, -1000, -2000

Nápověda:
• Cyklem (smyčkou LOOP) projdeme přes všechny položky pole.
• Aktuální položku pole zkopírujeme z pole do registru (load string),
vydělíme, a podíl vrátíme do aktuální položky pole (store string).

• Pozor na rozdíl v rozšiřování a dělení, se znaménkem a bez znaménka (viz. cv04).

Příznak směru
Směr posuvu u řetězových instrukcí určujeme příznakem směru (DF, Direction Flag):
• Příznak nastavujeme instrukcemi CLD (DF = 0, doprava) a STD (DF = 1, doleva).

1 cld ; CLear Direction flag (DF = 0, posun doprava)
2 std ; SeT Direction flag (DF = 1, posun doleva)

Pokud polem chceme projít ve směru zprava doleva, počáteční hodnotu indexových
indexových registrů ESI a EDI musíme nastavit na adresu jeho poslední položky:
• Poslední položka pole je v paměti posunutá o “(N-1) * velikost” bajtů:

3 mov esi, X ;ESI= X+0
4 cld ; DF= 0 (doprava)
5 lodsb ; AL= [X+0], ESI= X+1
6 lodsb ; AL= [X+1], ESI= X+2
7 lodsb ; AL= [X+2], ESI= X+3

8 mov esi, X+9;ESI= X+9
9 std ; DF= 1 (doleva)
10 lodsb ; AL= [X+9], ESI= X+8
11 lodsb ; AL= [X+8], ESI= X+7
12 lodsb ; AL= [X+7], ESI= X+6

Na vyzkoušení

Cyklus zleva doprava projde přes všechny (ECX) položky pole (ESI) a vypíše je:

%include 'rw32.inc' ;knihovna pro vstup a vystup

section .data ;inicializovany segment
arr dd 100,200,300,400,500 ; pole peti 32b polozek jmenem arr

section .text ;kodovy segment
main:

mov esi, arr ; ESI = adresa ciloveho pole (prvni prvek)

mov ecx, 5 ; ECX = pocet opakovani (N)
cld ; DF = 0 (posun zleva doprava)

.for:
lodsd ; zkopiruj 32b z adresy v ESI do EAX

; a ESI zvys (DF==0) o ctyrku (32b)
call WriteInt32NewLine ; vypis EAX
loop .for ; dekrementuj ECX, a pokud ECX != 0 tak skoc

ret

Na vyzkoušení

Cyklus zprava doleva projde přes všechny (ECX) položky pole (ESI) a vypíše je:

%include 'rw32.inc' ;knihovna pro vstup a vystup

section .data ;inicializovany segment
arr dd 100,200,300,400,500 ; pole peti 32b polozek jmenem arr

section .text ;kodovy segment
main:

mov esi, arr ; ESI = adresa ciloveho pole (prvni prvek)
add esi, 16 ; posun posledniho prvku "(N-1) * velikost"
mov ecx, 5 ; ECX = pocet opakovani (N)
std ; DF = 1 (posun zprava doleva)

.for:
lodsd ; zkopiruj 32b z adresy v ESI do EAX

; a ESI sniz (DF==1) o ctyrku (32b)
call WriteInt32NewLine ; vypis EAX
loop .for ; dekrementuj ECX, a pokud ECX != 0 tak skoc

ret

Řetězové porovnávání - CMPS a SCAS

Pole s polem porovnáme instrukcemi CMPSB, CMPSW, CMPSD (CoMPare String):
• Instrukce porovná data z adresy v ESI s daty na adrese v EDI,
a adresy v ESI a EDI posune na další položku.

1 cmpsb ;porovnej 8b z adresy v ESI a z adresy v EDI (a oba posun o 1B)
2 cmpsw ;porovnej 16b z adresy v ESI a z adresy v EDI (a oba posun o 2B)
3 cmpsd ;porovnej 32b z adresy v ESI a z adresy v EDI (a oba posun o 4B)

Registr s polem porovnáme instrukcemi SCASB, SCASW, SCASD (SCAn String):
• Instrukce porovná data z registru AL/AX/EAX s daty na adrese v EDI,
a adresu v EDI posune na další položku.

4 scasb ;porovnej 8b z AL a z adresy v EDI (a EDI posun o 1B)
5 scasw ;porovnej 16b z AX a z adresy v EDI (a EDI posun o 2B)
6 scasd ;porovnej 32b z EAX a z adresy v EDI (a EDI posun o 4B)

Obě řetězová porovnání příznaky nastavují stejně jako instrukce CMP (CoMPare):
• Po řetězovém porovnání tedy můžeme používat aritmetické skoky (viz. cv05).

Na vyzkoušení
Cyklus porovná řetězec (ESI) s řetězcem (EDI) a vypíše v kolika znacích se shodují:
%include 'rw32.inc' ;knihovna pro vstup a vystup
section .data ;inicializovany segment

src db "ABCAD", 0 ; zdrojovy retezec (pole 8b hodnot)
dst db "ABBAC", 0 ; cilovy retezec (pole 8b hodnot)

section .text ;kodovy segment
main:

mov esi, src ; ESI = adresa zdrojoveho pole
mov edi, dst ; EDI = adresa ciloveho pole
mov ecx, 5 ; ECX = pocet opakovani (delka retezce)
cld ; DF = 0 (posun doprava)
mov ebx, 0 ; EBX = pocitadlo shodnych znaku

.for:
cmpsb ; 8b porovnej znaky na adrese v ESI a v EDI

; a ESI i EDI zvys (DF==0) o jednicku (8b)
jne .skip ; pokud znaky nebyly stejne tak skoc
inc ebx ; jinak inkrementuj EBX

.skip:
loop .for ; dekrementuj ECX, a pokud ECX != 0 tak skoc
mov eax, ebx ; EAX = pocitadlo
call WriteInt32NewLine ; vypis EAX
ret

Na vyzkoušení
Cyklus porovná znak (AL) s řetězcem (EDI) a vypíše kolikrát se znak v řetězci vyskytuje:
%include 'rw32.inc' ;knihovna pro vstup a vystup
section .data ;inicializovany segment

dst db "ABBAC", 0 ; cilovy retezec (pole 8b hodnot)
section .text ;kodovy segment
main:

mov al, 'A' ; AL = hledany znak
mov edi, dst ; EDI = adresa ciloveho pole
mov ecx, 5 ; ECX = pocet opakovani (delka retezce)
cld ; DF = 0 (posun doprava)
mov ebx, 0 ; EBX = pocitadlo shodnych znaku

.for:
scasb ; 8b porovnej znak v AL a na adrese v EDI

; a EDI zvys (DF==0) o jednicku (8b)
jne .skip ; pokud znaky nebyly stejne tak skoc
inc ebx ; jinak inkrementuj EBX

.skip:
loop .for ; dekrementuj ECX, a pokud ECX != 0 tak skoc
mov eax, ebx ; EAX = pocitadlo
call WriteInt32NewLine ; vypis EAX
ret

Na procvičení
Vyzkoušejte si:
• Vytvořte si pole znaků a ze vstupu do něj načtěte textový řetězec.
• První výskyt znaku ‘ a ’ v řetězci nahraďte otazníkem ‘ ? ’.
• Poslední výskyt znaku ‘ b ’ v řetězci nahraďte vykřičníkem ‘ ! ’.
• Vypište upravený řetězec.

Například:
• abc => ?!c
• ccc => ccc
• aaabbbccc => ?aabb!ccc
• cccbbbaaa => cccbb!?aa

Nápověda:
• Řetězec načteme a vypíšeme funkcemi ReadString a WriteString (viz. cv03).
• Problém vyřešíme dvěma samostatnými cykly:
První cyklus jde z leva do prava a hledá (scan string) znak ‘ a ’.
Druhý cyklus jde z prava do leva a hledá (scan string) znak ‘ b ’.

• Když hledaný znak najdeme a nahradíme, tak ukončíme aktuální cyklus.

Šablona

%include 'rw32.inc' ;knihovna pro vstup a vystup

section .bss ;neinicializovany segment
arr resb 20 ; pole pro dvacet 8b hodnot (19 znaku a NUL)

section .text ;kodovy segment
main:

mov edi, arr ; EDI = adresa nacitaneho retezce
mov ebx, 19 ; EBX = maximalni pocet nacitanych znaku
call ReadStringNewLine ; ze vstupu nacteme retezec
mov ebx, eax ; EBX = pocet uspesne nactenych znaku

; ??? (prvni cyklus)

; ??? (druhy cyklus)

mov esi, arr ; ESI = aresa vypisovaneho retezce
mov ecx, ebx ; pocet vypisovanych znaku
call WriteStringNewLine ; vypis retezec
ret

Prefixy opakování



Prefix opakování - REP
Pro opakování řetězové instrukce můžeme místo cyklu použít prefix opakování:
• Prefix REP (REPeat) provede instrukci, dekrementuje ECX, a pokud ECX != 0,
tak instrukci bude opakovat.

Program zkopíruje pět (ECX) 8b položek z pole src (ESI) do pole dst (EDI):

%include 'rw32.inc' ;knihovna pro vstup a vystup
section .bss ;ne-inicializovany segment

dst resb 5 ; rezervujeme pole peti 8b hodnot
section .data ;inicializovany segment

src db 10,20,30,40,50 ; definujeme pole peti 8b hodnot
section .text ;kodovy segment
main:

mov esi, src ; ESI = adresa zdrojoveho pole
mov edi, dst ; EDI = adresa ciloveho pole
mov ecx, 5 ; ECX = pocet opakovani (pocet polozek pole)
cld ; DF = 0 (posun z leva do doprava)
rep movsb ; zkopiruj 8b z adresy v ESI na adresu v EDI

; a oba je zvys (posun doprava) o jednicku (8b)
; dekrementuj ECX, a pokud ECX != 0 tak opakuj

ret

Podmíněné prefixy opakování - REPE a REPNE

Prefix instrukce řetězového porovnání (CMPS, SCAS) může být podmíněný:
• Prefix REPE (REPeat while Equal) provede instrukci, dekrementuje ECX,
a pokud ECX != 0 a hodnoty se rovnaly, tak instrukci bude opakovat.

• Prefix REPNE (REPeat while Not Equal) provede instrukci, dekrementuje ECX,
a pokud ECX != 0 a hodnoty se nerovnaly, tak instrukci bude opakovat.

Výsledek řetězového porovnání poznáme podle hodnoty příznaku ZF a registru ECX:
• Pokud shoda byla tak ZF = 1 (skok JE), pokud nebyla tak ZF = 0 (skok JNE).
• Pokud se cyklus ukončil při posledním porovnání tak ECX = 0,
pokud se cyklus ukončil při před-posledním porovnání tak ECX = 1, atd.

1 repe cmpsd ; provnej polozky poli na adresach v ESI a v EDI
2 ; ESI i EDI zvys (pokud DF==0) nebo sniz (pokud DF==1)
3 ; o ctyrku (32b), dekrementuj ECX, a pokud se polozky
4 ; shodovaly a ECX != 0, tak instrukci opakuj
5 je .stejne: ; pokud (ZF==1) tak se vsechny polozky shodovaly
6 jne .odlisne: ; jinak (ZF==0) tak se nejake polozky odlisovaly
7 ; (pozici prvni odlisne polozky pozname dle hodnoty ECX)

Na vyzkoušení
Program v řetězci (EDI) vyhledá první výskyt (DF == 0) hledaného znaku (AL), a vypíše
jeho index, nebo index ukončujícího znaku NUL, pokud hledaný znak v řetězci není:
%include 'rw32.inc' ;knihovna pro vstup a vystup
section .data ;inicializovany segment

arr db "Ahoj", 0 ; prohledavany retezec (pole 8b hodnot)
section .text ;kodovy segment
main:

mov edi, arr ; EDI = prohledavany retezec
mov ecx, 5 ; ECX = pocet opakovani (delka retezce)
call ReadCharNewLine ; AL = hledany znak
cld ; DF = 0 (posun doprava)

repne scasb ; AL porovnej s aktualnim znakem na adrese
; v EDI, EDI zvys (DF==0) o jednicku (8b),
; dekrementuj ECX, a pokud se znaky nerovnaly
; a ECX != 0, tak instrukci opakuj

mov eax, 4 ; EAX = index znaku NUL
sub eax, ecx ; od EAX edecteme pocet zbyvajicich opakovani

; (pokud znak nebyl nalezen tak ECX = 0)
call WriteInt32NewLine ; vypis EAX (index nalezeneho znaku)
ret

